Modified GUPs

A clean worksheet for Calc17. And extended for corrected compations in Calc18, until Aug 29. 2014.

This worksheet is documented in a german-english fashion :) It covers the complete discussion starting from the T00-Integral solving until temperature calculations. In the later parts, most computations are done (or at least supported) by numerical predictions. Much code is not so well documented, just ask me for details.

-- Sven Koeppel @ FIAS.

```
(* Die neue Wahl *)
V[q_, n_] = 1 / (1 + q^(n + 2))

(* Die alte Wahl *)
(*V[q_, n_] = 1/(1 + q^2)*)
vSide[q_, n_, side_] := q^(1 + n) * If[side > 0, V[q, n], (-1)^n V[-q, n]]
vEff[q_, n_] = vSide[q, n, +1] HeavisideTheta[q] + vSide[q, n, -1] HeavisideTheta[-q]
```

Out[1]= $$\frac{1}{1+q^{2+n}}$$

Out[3]= $$\frac{(-1)^n q^{1+n} \,\text{HeavisideTheta}[-q]}{1+(-q)^{2+n}}+\frac{q^{1+n}\,\text{HeavisideTheta}[q]}{1+q^{2+n}}$$

In[4]:=
```
Plot[Table[vEff[q, n], {n, 0, 7}] // Evaluate,
 {q, -3, 3},
 PlotStyle → (Riffle[#, {Dashed, #} & /@ Reverse[#]] & [ColorData[9, "ColorList"]]),
 PlotLegends → LineLegend[Table[n, {n, 0, 7}], LegendLabel → "n="],
 PlotRange → {-1, 1},
 ImageSize → Medium]
```

Out[4]=



Polstellen extrahieren

Code zwar zusammenkopiert, aber drübergeschaut.

In[5]:=
```
ReduceToSolutions[reduceResult_, extractVariable_] :=
 Cases[reduceResult, Equal[extractVariable, value_] → value, 100]
polReduce[n_, side_] := Reduce[Denominator@vSide[q, n, side] == 0, q]
polstellenSide[n_, side_] := polReduce[n, side] ~ ReduceToSolutions ~ q
polstellen[n_] := Join @@ (Select[polstellenSide[n, #[[1]]], #[[2]]] &) /@ {
     +1 →  (Re[#] ≥ 0 &),
     -1 → (Re[#] ≤ 0 &)
    }
mitgenommenePolstellen[n_] := Select[polstellen[n], Im[#] ≥ 0 &]
```

In[10]:=
```
(* Zeichne die Polstellen.
  Diese Funktion/Zelle ist aus Polstellen.nb kopiert und etwas modifiziert. *)
Daten[n_] := {
  {Re[#], Im[#]} & /@ (Solve[z^(2+n) == 1, z] /. {{z → a_} → a}) , (* in Blau *)
  {Re[#], Im[#]} & /@ polstellenSide[n, +1],
  {Re[#], Im[#]} & /@ polstellenSide[n, -1],
  {Re[#], Im[#]} & /@ polstellen[n] (* schwarz *),
  {Re[#], Im[#]} & /@ mitgenommenePolstellen[n]
 }
MakePlot[n_,
   label_: "Left and right hand poles"
  ] := Show[ListPlot[
    Daten[n],
    AxesOrigin → {0, 0},
    PlotRange → {{-1.2, 1.2}, {-1.2, 1.2}},
    AspectRatio → 1,
    Frame → True,
    FrameLabel → {{Im, None}, {Re, label }},
    PlotMarkers → {
      {Graphics[{Lighter@Blue, Opacity[0.3], Disk[]}], 0.05},
      {
       Graphics@{Darker@Yellow,
         Circle[],
         Disk[{0, 0}, 1, {-Pi/2, Pi/2}],
         Text[
          Style["R", "Title", 15, Bold, Darker@Darker@Brown], {+0.5, Center}]},
       0.1},
      {
       Graphics@{Lighter@Magenta,
         Circle[],
         Disk[{0, 0}, 1, {Pi/2, 3/2*Pi}],
         Text[Style["L", "Title", 15, Bold, Darker[Magenta]], {-0.5, Center}]},
       0.1},
      {Graphics[{Black, Thickness[0.15], Circle[]}], 0.12},
      {Graphics[{Red, Thickness[0.15], Circle[]}], 0.18}
     },
    PlotRange → All,
    ImageSize → Medium],
   Graphics[{Opacity[0.2], Circle[{0, 0}, 1]}]
  ];
Manipulate[Show[
  MakePlot[n],
  Graphics[
   {Text[Style[StringForm["n=``", n], {Black, 20}], Offset[{120, 130}, {0, 0}]](*,
    Text[Style[StringForm["# Residuen=``",ResidueList[n]//Length],{Black, 20}],
     Offset[{+30,-30},{0,0}]]*)}
  ]]
 , {n, 0, 15, 1}]
```

Out[12]=



In[13]:=
```
Grid[Join @@ { {Style[#, {Blue, Bold, 12}] & /@
     {"n", "# unique poles", "# poles", "Unique Poles"}},
   Table[{n,
     Length@Union@polstellen[n],
     Length@polstellen[n],
     Union@polstellen[n] // Sort
    }, {n, 0, 7}]},
 Frame → All, Alignment → {Left, Left}]
```

Out[13]=

n	# unique poles	# poles	Unique Poles
0	2	4	$\{-i, i\}$
1	4	4	$\{-(-1)^{1/3}, (-1)^{1/3}, -(-1)^{2/3}, (-1)^{2/3}\}$
2	4	4	$\{-(-1)^{1/4}, (-1)^{1/4}, -(-1)^{3/4}, (-1)^{3/4}\}$
3	4	4	$\{-(-1)^{1/5}, (-1)^{1/5}, -(-1)^{4/5}, (-1)^{4/5}\}$
4	6	8	$\{-i, i, -(-1)^{1/6}, (-1)^{1/6}, -(-1)^{5/6}, (-1)^{5/6}\}$
5	8	8	$\{-(-1)^{1/7}, (-1)^{1/7}, -(-1)^{3/7}, (-1)^{3/7}, -(-1)^{4/7}, (-1)^{4/7}, -(-1)^{6/7}, (-1)^{6/7}\}$
6	8	8	$\{-(-1)^{1/8}, (-1)^{1/8}, -(-1)^{3/8}, (-1)^{3/8}, -(-1)^{5/8}, (-1)^{5/8}, -(-1)^{7/8}, (-1)^{7/8}\}$
7	8	8	$\{-(-1)^{1/9}, (-1)^{1/9}, -(-1)^{1/3}, (-1)^{1/3}, -(-1)^{2/3}, (-1)^{2/3}, -(-1)^{8/9}, (-1)^{8/9}\}$

In[14]:=
```
Grid[Join @@ {{Style[#, {Blue, Bold, 12}] & /@
      {"n", "interesting poles"}},
    Table[{n,
       Length@Select[Arg@mitgenommenePolstellen[n], # ≤ π/2 &],
       Select[Arg@mitgenommenePolstellen[n], # ≤ π/2 &]
      }, {n, 0, 7}]},
  Frame → All, Alignment → {Left, Left}]
```

Out[14]=

n	interesting poles	
0	2	$\{\frac{\pi}{2}, \frac{\pi}{2}\}$
1	1	$\{\frac{\pi}{3}\}$
2	1	$\{\frac{\pi}{4}\}$
3	1	$\{\frac{\pi}{5}\}$
4	3	$\{\frac{\pi}{2}, \frac{\pi}{6}, \frac{\pi}{2}\}$
5	2	$\{\frac{\pi}{7}, \frac{3\pi}{7}\}$
6	2	$\{\frac{\pi}{8}, \frac{3\pi}{8}\}$
7	2	$\{\frac{\pi}{9}, \frac{\pi}{3}\}$

In[15]:=

Integralberechnung

Achtung wegen der Vorfaktoren:

* Die Integralresiduenliste beinhaltet die kompletten Residuen, also auch den res $f_\pm = \frac{1}{2+n}$-Fakt.

* AValue beinhaltet den $\frac{1}{-2iz}$-Faktor und damit einen Teil von f_0. Grund: AValue[0] wird so sehr einfach.

In[16]:=
```
IntegralResidueList[n_] :=
  (2 Pi I) * (+1) * Residue[vSide[q, n, Sign@Re@#] * Exp[+I q z], {q, #}] & /@
   mitgenommenePolstellen[n]
IntegralValue[n_] := Total@IntegralResidueList[n]
AValue[n_] := - I / (2 z) * IntegralValue[n]
```

In[19]:=
```
Grid[Table[{
   n,
   Grid[{mitgenommenePolstellen[n]}, Frame → All],
   Grid[{IntegralResidueList[n] / (2 Pi I)}, Frame → All]
  }, {n, 0, 7}], Frame → All, Alignment → {Left, Left}]
```

Out[19]=

0	$\mathbb{i}$ $\mathbb{i}$	$\frac{e^{-z}}{2}$ $\frac{e^{-z}}{2}$
1	$(-1)^{1/3}$ $(-1)^{2/3}$	$\frac{1}{3} e^{(-1)^{5/6} z}$ $\frac{1}{3} e^{-(-1)^{1/6} z}$
2	$(-1)^{1/4}$ $(-1)^{3/4}$	$\frac{1}{4} e^{(-1)^{3/4} z}$ $\frac{1}{4} e^{-(-1)^{1/4} z}$
3	$(-1)^{1/5}$ $(-1)^{4/5}$	$\frac{1}{5} e^{(-1)^{7/10} z}$ $\frac{1}{5} e^{-(-1)^{3/10} z}$
4	$\mathbb{i}$ $(-1)^{1/6}$ $\mathbb{i}$ $(-1)^{5/6}$	$\frac{e^{-z}}{6}$ $\frac{1}{6} e^{(-1)^{2/3} z}$ $\frac{e^{-z}}{6}$ $\frac{1}{6} e^{-(-1)^{1/3} z}$
5	$(-1)^{1/7}$ $(-1)^{3/7}$ $(-1)^{4/7}$ $(-1)^{6/7}$	$\frac{1}{7} e^{(-1)^{9/14} z}$ $\frac{1}{7} e^{(-1)^{13/14} z}$ $\frac{1}{7} e^{-(-1)^{1/14} z}$ $\frac{1}{7} e^{-(-1)^{5/14} z}$
6	$(-1)^{1/8}$ $(-1)^{3/8}$ $(-1)^{5/8}$ $(-1)^{7/8}$	$\frac{1}{8} e^{(-1)^{5/8} z}$ $\frac{1}{8} e^{(-1)^{7/8} z}$ $\frac{1}{8} e^{-(-1)^{1/8} z}$ $\frac{1}{8} e^{-(-1)^{3/8} z}$
7	$(-1)^{1/9}$ $(-1)^{1/3}$ $(-1)^{2/3}$ $(-1)^{8/9}$	$\frac{1}{9} e^{(-1)^{11/18} z}$ $\frac{1}{9} e^{(-1)^{5/6} z}$ $\frac{1}{9} e^{-(-1)^{1/6} z}$ $\frac{1}{9} e^{-(-1)^{7/18} z}$

In[20]:=
```
Grid[Join @@ { {Style[#, {Blue, Bold, 12}] & /@
     {"n", "# poles", "Wert für A(p)"}},
   Table[{n,
     Length@mitgenommenePolstellen[n],
     AValue[n] ,
     AValue[n] // ComplexExpand
     (*AValue[n]//ComplexExpand // FullSimplify*)
    }, {n, 0, 8}]},
 Frame → All, Alignment → {Left, Left}]
```

Out[20]=

n	# poles	Wert für A(p)	
0	2	$\frac{e^{-z}\pi}{z}$	$\frac{e^{-z}\pi}{z}$
1	2	$-\frac{i\left(\frac{2}{3}\, i\, e^{-(-1)^{1/6} z}\pi+\frac{2}{3}\, i\, e^{(-1)^{5/6} z}\pi\right)}{2 z}$	$\frac{2\, e^{-\frac{\sqrt{3}\, z}{2}}\pi \operatorname{Cos}\left[\frac{z}{2}\right]}{3 z}$
2	2	$-\frac{i\left(\frac{1}{2}\, i\, e^{-(-1)^{1/4} z}\pi+\frac{1}{2}\, i\, e^{(-1)^{3/4} z}\pi\right)}{2 z}$	$\frac{e^{-\frac{z}{\sqrt{2}}}\pi \operatorname{Cos}\left[\frac{z}{\sqrt{2}}\right]}{2 z}$
3	2	$-\frac{i\left(\frac{2}{5}\, i\, e^{-(-1)^{3/10} z}\pi+\frac{2}{5}\, i\, e^{(-1)^{7/10} z}\pi\right)}{2 z}$	$\frac{e^{-\sqrt{\frac{5}{8}-\frac{\sqrt{5}}{8}}\, z}\pi \operatorname{Cos}\left[\frac{1}{4}\left(-1-\sqrt{5}\right) z\right]}{5 z}+\frac{e^{-\sqrt{\frac{5}{8}-\frac{\sqrt{5}}{8}}\, z}\pi \operatorname{Cos}\left[\frac{1}{4}\left(1+\sqrt{5}\right) z\right]}{5 z}+i\left(\frac{e^{-\sqrt{\frac{5}{8}-\frac{\sqrt{5}}{8}}\, z}\pi \operatorname{Sin}\left[\frac{1}{4}\left(-1-\sqrt{5}\right) z\right]}{5 z}+\frac{e^{-\sqrt{\frac{5}{8}-\frac{\sqrt{5}}{8}}\, z}\pi \operatorname{Sin}\left[\frac{1}{4}\left(1+\sqrt{5}\right) z\right]}{5 z}\right)$
4	4	$-\frac{i\left(\frac{2}{3}\, i\, e^{-z}\pi+\frac{1}{3}\, i\, e^{-(-1)^{1/3} z}\pi+\frac{1}{3}\, i\, e^{(-1)^{2/3} z}\pi\right)}{2 z}$	$\frac{e^{-z}\pi}{3 z}+\frac{e^{-z/2}\pi \operatorname{Cos}\left[\frac{\sqrt{3}\, z}{2}\right]}{3 z}$
5	4	$-\frac{1}{2 z} i\left(\frac{2}{7}\, i\, e^{-(-1)^{1/14} z}\pi+\frac{2}{7}\, i\, e^{-(-1)^{5/14} z}\pi+\frac{2}{7}\, i\, e^{(-1)^{9/14} z}\pi+\frac{2}{7}\, i\, e^{(-1)^{13/14} z}\pi\right)$	$\frac{2\, e^{-z \operatorname{Sin}\left[\frac{\pi}{7}\right]}\pi \operatorname{Cos}\left[z \operatorname{Cos}\left[\frac{\pi}{7}\right]\right]}{7 z}+\frac{2\, e^{-z \operatorname{Cos}\left[\frac{\pi}{14}\right]}\pi \operatorname{Cos}\left[z \operatorname{Sin}\left[\frac{\pi}{14}\right]\right]}{7 z}$
6	4	$-\frac{1}{2 z} i\left(\frac{1}{4}\, i\, e^{-(-1)^{1/8} z}\pi+\frac{1}{4}\, i\, e^{-(-1)^{3/8} z}\pi+\frac{1}{4}\, i\, e^{(-1)^{5/8} z}\pi+\frac{1}{4}\, i\, e^{(-1)^{7/8} z}\pi\right)$	$\frac{e^{-z \operatorname{Sin}\left[\frac{\pi}{8}\right]}\pi \operatorname{Cos}\left[z \operatorname{Cos}\left[\frac{\pi}{8}\right]\right]}{4 z}+\frac{e^{-z \operatorname{Cos}\left[\frac{\pi}{8}\right]}\pi \operatorname{Cos}\left[z \operatorname{Sin}\left[\frac{\pi}{8}\right]\right]}{4 z}$
7	4	$-\frac{1}{2 z} i\left(\frac{2}{9}\, i\, e^{-(-1)^{1/6} z}\pi+\frac{2}{9}\, i\, e^{-(-1)^{7/18} z}\pi+\frac{2}{9}\, i\, e^{(-1)^{11/18} z}\pi+\frac{2}{9}\, i\, e^{(-1)^{5/6} z}\pi\right)$	$\frac{2\, e^{-\frac{\sqrt{3}\, z}{2}}\pi \operatorname{Cos}\left[\frac{z}{2}\right]}{9 z}+\frac{2\, e^{-z \operatorname{Sin}\left[\frac{\pi}{9}\right]}\pi \operatorname{Cos}\left[z \operatorname{Cos}\left[\frac{\pi}{9}\right]\right]}{9 z}$
8	6	$-\frac{1}{2 z} i\left(\frac{2}{5}\, i\, e^{-z}\pi+\frac{1}{5}\, i\, e^{-(-1)^{1/5} z}\pi+\frac{1}{5}\, i\, e^{-(-1)^{2/5} z}\pi+\frac{1}{5}\, i\, e^{(-1)^{3/5} z}\pi+\frac{1}{5}\, i\, e^{(-1)^{4/5} z}\pi\right)$	$\frac{e^{-z}\pi}{5 z}+\frac{e^{\frac{1}{4}\left(-1-\sqrt{5}\right) z}\pi \operatorname{Cos}\left[\sqrt{\frac{5}{8}-\frac{\sqrt{5}}{8}}\, z\right]}{5 z}+\frac{e^{\frac{1}{4}\left(1-\sqrt{5}\right) z}\pi \operatorname{Cos}\left[\sqrt{\frac{5}{8}+\frac{\sqrt{5}}{8}}\, z\right]}{5 z}$

Für das korrekte Ergebnis $\vartheta_0 = f_0\, A_{\text{value}}$ beinhaltet das f_0 in diesem Worksheet die restlichen Faktoren im Vergleich zum Paper

In[72]:=
```
Ω[d_] := 2 Pi^((d+1)/2) / Gamma[(d+1)/2]
f0[n_] := Ω[n + 2] / (2 π)^(2+n) (* Echte Vorfaktoren,
oben in AValue schon was Teile mitgenommen *)
Plot[Evaluate@Table[f0@n AValue@n, {n, 0, 2}], {z, 0.1, 1},
 PlotStyle → (Riffle[#, {Dashed, #} & /@ Reverse[#]] & [ColorData[9, "ColorList"]]),
 PlotLegends → Placed[LineLegend[Table[n, {n, 0, 7}],
    LegendLabel → "n=", LegendLayout → {"Row", 3}], {Right, Bottom}],
 FrameLabel → {"r / L*", "T00 / M"},
 PlotLabel → "Energy Density T00 of modified GUP",
 Frame → True(*,
 PlotRange→ {-1,1}*)
]
```

Out[74]=


Zur Berechnung von g_{00} nutzen wir generische Lösung der Rizzo-DGL. Dabei wird an dieser Stelle hier das M reingemogelt, was dem physikalischen $M/M^{n+2}{}_*$ entspricht.

In[330]:=
```
g00[n_, r_] := 1 - M/r^(1+n) 2/(n + 2) Integrate[z^(n+2) f0[n] AValue[n], {z, 0, r}]
g00s = Table[g00[n, r], {n, 0, 7}];
```

Looking for the extremal radius

In[291]:=
```
(* Prepare Export and metric units *)
SetDirectory[NotebookDirectory[]];
cm = 72 / 2.54;
```

In[598]:=
```
(* Epic Epilog-Labels coming here, modified for the current problem:
     Featuring scaling and correct Inset (no rectangle nonrotated box).
     ASSUMING data be a list of X as variable, ideal for g00s. *)
ClearAll[EpiLabels]
EpiLabels[xpoint_, data_, labels_, axisratio_: 1] :=
 With[{xpoints = If[Head@xpoint === List, xpoint, xpoint & /@ labels]},
  (* erlaube Liste von xpoints *)
  Inset[Rotate[Text[labels[[#]], Background → White],
      ArcTan@Re@N[(D[data⟦#⟧, X] /. {X → xpoints⟦#⟧}) ] / axisratio] ,
     {xpoints⟦#⟧, 0 + ( data /. X → xpoints⟦#⟧) ⟦#⟧}
    ] & /@ Range@Length@data] (* 1: any value to introspect list*)

point[color_, coord_, size_: 0.37] := Inset[
  Graphics@{Opacity[0.9], color, Disk[]},
  (* {x,y}, {xOrig,yOrig}, scaling --> see Inset doku *)
  coord, {0, 0}, size
 ]
miniToSol[res_] := res /. {y_, {_ → x_}} → {x, y}
(* for FindMinimum → Coordinate  *)
rootToSol[res_] := res /. {_ → x_} → x (* for first FindRoot → value *)
```

Find r_C and M_* numerically

In[504]:=
```
(* find extremal r0 Values. The test Masses are "experimental" *)
r0s = First@Transpose@Table[miniToSol@FindMinimum[
        Re@g00s⟦n + 1⟧ /. M → 10 (n + 1)^6, {r, 1}
        ], {n, 0, Length@g00s - 1}];
(* fi3d extremal Mass values *)
Ms = Table[rootToSol@FindRoot[Re@g00s⟦n + 1⟧ /. r → r0s⟦n + 1⟧, {M, 10^n}] ,
   {n, 0, Length@g00s - 1}];
Grid[{
  {"n"}~Join~Range[0, Length@g00s - 1],
  {"r0"}~Join~r0s,
  {"M*"}~Join~Ms
 }, Frame → All]
```

FindMinimum::sdprec :

Line search unable to find a sufficient decrease in the function value with MachinePrecision digit precision. ≫

Out[506]=

n	0	1	2	3	4	5	6	7
r_0	1.79328	1.27534	1.07714	0.993701	1.01592	0.981144	0.953649	0.932502
M_*	3.35092	53.0073	621.491	6536.7	35 182.9	359 680.	3.69058×10^6	3.8323×10^7

In[507]:= **% // TeXForm**

Out[507]//TeXForm=

```
\begin{array}{ccccccccc}
 \text{n} & 0 & 1 & 2 & 3 & 4 & 5 & 6 & 7 \\
 r_0 & 1.79328 & 1.27534 & 1.07714 & 0.993701 & 1.01592 & 0.981144 & 0.953649 & 0.93
 M_* & 3.35092 & 53.0073 & 621.491 & 6536.7 & 35182.9 & 359680. & 3.69058\times 10^6
   3.8323\times 10^7 \\
\end{array}
```

Interactive Plot fuer variierende M, n.

In[545]:=
```
getM[n_] := Ms[[n + 1]]
Manipulate[
 Plot[Evaluate@Re@({g00s[[n + 1]], 1 - 2 M/((n + 2) Ω[n]) 1/r^(1+n)} /. {M → m}),
  {r, 0.03, 10},
  PlotRangePadding → 0,
  PlotStyle → {Red, Blue},
  Frame → True,
  FrameLabel → {"r / L", "g00"},
  PlotRange → {-1, 1.7},
  GridLines → {{r0s[[n + 1]]}, {1}},
  ImageSize → Medium,
  Epilog → {point[Red, {r0s[[n + 1]], g00s[[n + 1]] /. {r → r0s[[n + 1]], M → m} }]},
  PlotLabel → "Metric g00 of modified GUP in n=" <> ToString@n <>
    " LXDs\n r0=" <> ToString@r0s[[n + 1]] <> ", g00(r0)=" <> ToString@Ms[[n + 1]]
    <> ", M=" <> ToString@m],
 {{m, 1}, 0.4, 2 getM[n]}, {n, 0, Length@g00s - 1, 1}]
```

Out[546]=



In[496]:=
```
(* @Re@: Cut off small imaginary parts *)
fig = Module[{n = 2, chosenM},
  chosenM = {250, Ms[[n + 1]], 1000};
  pointSize = 0.15;
  labels = "M=" <> ToString@# <> "M*" & /@ chosenM ;
  labelsSMM = "STM, " <> # & /@ labels;
  SMM = 1 - (2 M)/(n + 2) 1/r^(1+n);
  Plot[Evaluate@Re@Flatten@Table[
      {g00s[[n + 1]], SMM} /. {M → M}, {M, chosenM}], {r, 0, 10},
   PlotRangePadding → 0,
   PlotStyle →
    Flatten[{#, {Dashed, Lighter@#}} & /@ {Darker@Green, Red, Blue}, 1],
   Frame → True,
   FrameLabel → {"r / L", "g00"},
   PlotRange → {-0.7, 1.7},
   Epilog → {
     EpiLabels[6, g00s[[n + 1]] /. {M → #, r → X} & /@ chosenM, labels, 5 / 10],
     EpiLabels[8.5, SMM /. {M → #,  r → X} & /@ chosenM, labelsSMM, 4.1 / 10],
     point[Red, {r0s[[n + 1]], 0}, pointSize],
     point[Blue, {rootToSol@
         FindRoot[Re@g00s[[n + 1]] /. M → Last@chosenM, {r, 1}], 0}, pointSize],
     point[Blue, {rootToSol@FindRoot[Re@g00s[[n + 1]] /. M → Last@chosenM, {r, 2}],
       0}, pointSize]
    },
   GridLines → {{r0s[[n + 1]]}, {1}},
   ImageSize → 15 cm,
   AspectRatio → 0.6
   (* wird in Latex gemacht: *)
   (*lotLabel→ "Metric component g00 of modified GUP in n=2 LXDs"]*)
  ]]
```

Out[496]=



```
In[497]:= Export["../../Master-Calc18/figures/g00-n2.pdf", fig]

Out[497]= ../../Master-Calc18/figures/g00-n2.pdf
```

```
fig = Module[{n = 5, chosenM},
  chosenM = {0.5 Ms[[n + 1]], Ms[[n + 1]], 1.5 Ms[[n + 1]]};
  pointSize = 0.15;
  labels = {"M=0.5 M*", "M=M*", "M=1.5 M*"};
  SMM = 1 - (2 M)/(n + 2) 1/r^(1+n);
  Plot[Evaluate@Re@Flatten@Table[
       {g00s[[n + 1]], SMM} /. {M → M}, {M, chosenM}], {r, 0, 10},
   PlotRangePadding → 0,
   PlotStyle →
    Flatten[{#, {Dashed, Lighter@#}} & /@ {Darker@Green, Red, Blue}, 1],
   Frame → True,
   FrameLabel → {"r / L", "g00"},
   PlotRange → {-0.7, 1.7},
   Epilog → {
     EpiLabels[{4.5, 5.0, 5.5},
      g00s[[n + 1]] /. {M → #, r → X} & /@ chosenM, labels, 6 / 10],
     point[Red, {r0s[[n + 1]], 0}, pointSize],
     point[Blue, {rootToSol@
         FindRoot[Re@g00s[[n + 1]] /. M → Last@chosenM, {r, 0.5}], 0}, pointSize],
     point[Blue, {rootToSol@FindRoot[Re@g00s[[n + 1]] /. M → Last@chosenM, {r, 2}],
       0}, pointSize]
    },
   GridLines → {{r0s[[n + 1]]}, {1}},
   Exclusions → Range[0, 0.1], (* hat quasi keinen Effekt...
     sollte bloss senkrechte Striche bei r=0 unterbinden *)
   ImageSize → 15 cm,
   AspectRatio → 0.6
   (* Wird in Latex gemacht *)
   (*PlotLabel→
     "Metric component g00 of modified GUP in n="<>ToString@n<>" LXDs"]*)
  ]]
```

Out[502]=



In[503]:=
```
Export["../../Master-Calc18/figures/g00-n5.pdf", fig]
```

Out[503]=
```
../../Master-Calc18/figures/g00-n5.pdf
```

Temperature

In[589]:=
```
(* wg Simplify ca 10sec Laufzeit *)
Ts = 1 / (4 π M) D[g00s, r] * Table[Ω[n + 1], {n, 0, Length@g00s - 1}];
```

In[590]:=
```
Off[FindMaximum::lstol]
TsMaxima = miniToSol@FindMaximum[Re@#, {r, 2}] & /@ Ts
```

Out[591]=
$\{\{3.38363, 0.0202156\}, \{2.51257, 0.00655127\}, \{2.26118, 0.00166479\}, \{2.18714, 0.000309357\}, \{2.19159, 0.0000558368\}, \{2.17113, 6.97503\times10^{-6}\}, \{2.1577, 7.76522\times10^{-7}\}, \{2.14889, 7.74554\times10^{-8}\}\}$

In[676]:=
```
fig = Module[{chosenN},
  chosenN = {0, 1, 2};
  colors = {Darker@Green, Darker@Yellow, Red};
  labels = "n=" <> ToString@# & /@ chosenN;
  TData = Table[Ts[[n + 1]] /. {r → r * r0s[[n + 1]]}, {n, chosenN}];
  TSSM = Table[D[1 - 2 / (r^(1+n)), r] /. {r → r r0s[[n + 1]]}, {n, chosenN}];
  Plot[Evaluate@Join@{Re@TData, TSSM},
   {r, 0, 10},
   PlotStyle → Join[colors, {Dashed, Lighter@#} & /@ colors],
   PlotRangePadding → None,
   FrameLabel → {"r_H / r_0", "T/L_P"},
   Frame → True,
   Epilog → Join[
     {EpiLabels[3.4, TData /. r → X, labels , 0.02 / 3]},
     {EpiLabels[{6, 5, 4.5}, TSSM /. r → X, labels , 0.02 / 2.5]},
     Inset[
        Graphics@{Opacity[0.9], #[[2]], Disk[]},
        (* {x,y}, {xOrig,yOrig}, scaling --> see Inset doku *)
        {#[[1, 1]] / #[[3]], #[[1, 2]]}, {0, 0}, 0.17
       ] & /@ Transpose@{TsMaxima[[chosenN + 1]], colors, r0s[[chosenN + 1]]}
    ],
   PlotRange →
    {-0.1 Max@Last@Transpose@TsMaxima, 1.1 Max@Last@Transpose@TsMaxima},
   GridLines → {{}, {0}},
   AspectRatio → 0.5,
   ImageSize → 15 cm
  ]
 ]
```

Out[676]=



In[677]:=
```
Export["../../Master-Calc18/figures/Temp.pdf", fig]
```

Out[677]=
```
../../Master-Calc18/figures/Temp.pdf
```

In[697]:=
```
fig = Module[{chosenN},
  chosenN = {7};
  colors = {Darker@Green};
  labels = "n=" <> ToString@# & /@ chosenN;
  TData = Table[Ts⟦n + 1⟧ /. {r → r * r0s⟦n + 1⟧}, {n, chosenN}];
  TSSM = Table[D[1 - 2 / (r^(1+n)), r] /. {r → r r0s⟦n + 1⟧}, {n, chosenN}];
  Plot[Evaluate@Join@{Re@TData, TSSM},
   {r, 0.4, 15},
   PlotStyle → Join[colors, {Dashed, Lighter@#} & /@ colors],
   PlotRangePadding → None,
   FrameLabel → {"r_H / r_0", "T/L_P"},
   Frame → True,
   PlotRange → {-1.1 Max@Last@Transpose@TsMaxima⟦chosenN + 1⟧,
     1.1 Max@Last@Transpose@TsMaxima⟦chosenN + 1⟧},
   GridLines → {{}, {0}},
   AspectRatio → 0.3,
   Exclusions → Range[0, 0.1],
   ImageSize → 15 cm
  ]
 ]
```

Out[697]=



In[693]:=
```
Export["../../Master-Calc18/figures/Temp-n7.pdf", fig]
```

Out[693]=
```
../../Master-Calc18/figures/Temp-n7.pdf
```